Predicting Dam-break Flood Depths using a One-Dimensional Diffusion Model

T.V. Hromadka II

Hydrologist, U.S. Geological Survey, Laguna Niguel, California,

ABSTRACT

A one-dimensional diffusion model of unsteady flow in open channels is used to estimate the flooding depths resulting from a hypothetical failure of an earthen dam. In order to evaluate the accuracy achieved by the diffusion model in the prediction of flood depths, a well known one-dimensional model, which solves the complete coupled flow equations of momentum and continuity (K-634), is used to develop flood depths for comparison purposes. For the dam-break hydrographs and range of downstream channel slopes considered, it is concluded that the simpler diffusion modeling approach produces flood-depth estimates which are comparable to the results produced from the more complex K-634 model. The study results suggest that the diffusion modeling approach can be extended to the development of a two-dimensional diffusion model for the estimation of two-dimensional dam-break flood plains.

KEY WORDS

Dam-Break Models; Numerical Modeling; Finite-Element Method.

INTRODUCTION

The use of numerical methods to approximately solve the flow equations for the propagation of a flood wave due to an earthen dam failure has been the subject of several recent studies. Generally, the one-dimensional flow is modeled wherever there is no significant lateral variation in the flow. Land [7,8] examines four such dam-break models in their prediction of flooding levels and flood wave travel time, and compares the results against observed dam failure information. In dam-break analysis studies, an assumed dam-break failure mode (which may be part of the solution) is used to develop an inflow hydrograph to the downstream flood plain. Consequently, it is noted that a considerable sensitivity in modeling results is attributed to the dam-break failure rate assumptions. Ponce and Tsivoglou [13] examine the gradual failure of an earthen embankment (caused by an overtopping flooding event) and present a detailed model of the total system: sediment transport, unsteady channel hydraulics, and earth embankment failure. In this paper, the main objective is to evaluate the diffusion form of the flow equations for the estimation of flood depths (and the flood plain) resulting from a specified dam-break hydrograph. Consequently, the dam-break failure mode is not considered in this study. Rather, the dam-break failure mode may be included as part of the model solution (such as for a sudden breach) or specified as a reservoir outflow hydrograph.

In another study, Rajar [14] studies a one-dimensional flood wave propagation from an earthen dam failure. His model solves the St. Venant equations by means of either a first-order diffusive or a second-order Wendroff numerical scheme. A review of the literature indicates that the most often used numerical scheme is the method of characteristics (to solve the governing flow equations) such as described in Sakkas and Strelkoff [15] and Chen [2,3].

Although many dam-break studies involve flood flow regimes which are truly three-dimensional (in the horizontal plane), the two-dimensional case has not received much attention in the literature. Katopodes and Strelkoff [6] use the method of bicharacteristics to solve the governing equations of continuity and momentum. The model utilizes a moving grid algorithm to follow the flood wave propagation, and also employs several interpolation schemes to approximate the nonlinearity effects. In a much simpler approach, Xanthopoulos and Koutitas [16] use a diffusion model (i.e., the inertia terms are assumed negligible in a comparison to the pressure, friction, and gravity components) to approximate a two-dimensional flow field. The model assumes that the flood plain flow regime is such that the inertia terms (local and convective acceleration) are negligible. In a one-dimensional model, Akan and Yen [1] also use the diffusion approach to model hydrograph confluences at channel junctions. In the latter study, comparisons of modeling results were made between the diffusion model, a complete dynamic wave model solving the total equation system, and the basic kinematic wave equation model (i.e., the inertia and pressure terms are assumed negligible in comparison to the friction and gravity terms). The comparisons between the diffusion model and the dynamic wave model were very favorable, showing only minor discrepancies.

The kinematic-wave flow model has been recently used in the computation of dam-break flood waves [5]. Hunt concludes in his study that the kinematic-wave solution is asymptotically valid. Since the diffusion model has a wider range of applicability of bed slopes and wave periods than the kinematic model [12], then the diffusion model approach should provide an extension to the referenced kinematic model.

Because the diffusion modeling approach leads to an economic two-dimensional dam-break model (with numerical solutions based on the usual integrated finite-difference or finite element techniques), the need to include the extra components in the momentum equation must be studied. For example, evaluating the convective acceleration terms in a two-dimensional flow model requires approximately an additional 50-percent of the computational effort required in solving the entire two-dimensional model with the inertia terms omitted. Consequently, including the local and convective acceleration terms increases the computer execution costs significantly. Such increases in computational effort may not be significant for one-dimensional case studies; however, two-dimensional case studies necessarily involve considerably more computational effort and any justifiable simplifications of the governing flow equations is reflected by a significant decrease in computer software requirements, costs and computer execution time.

Ponce [11] examines the mathematical expressions of the flow equations which lead to wave attenuation in prismatic channels. It is concluded that the wave attenuation process is caused by the interaction of the local acceleration term with the sum of the terms of friction slope and channel slope. When local acceleration is considered negligible, wave attenuation is caused by the interaction of the friction slope and channel slope terms with the pressure gradient or convective acceleration terms (or a combination of both terms). Other discussions of flow conditions and the sensitivity to the various terms of the flow equations are given in Miller and Cunge [9], Morris and Woolhiser [10], and Henderson [4].

It is stressed that the ultimate objective of this research is to develop a two-dimensional diffusion model for use in estimating dam-break flood plains. Prior to finalizing such a model, however, the requirement of including the inertia terms in the unsteady flow equations needs to be ascertained.

The strategy used to check on this requirement is to evaluate the accuracy in predicted flood depths produced from a one-dimensional diffusion model with respect to the one-dimensional K-634 dam-break model which includes all of the inertia term components.

DIFFUSION MODEL

The mathematical relationships used in a one-dimensional diffusion model dam-break formulation are based upon the flow equations of continuity and momentum [1].

$$\frac{\partial Q_x}{\partial x} + \frac{\partial A_x}{\partial t} = 0 \tag{1}$$

$$\frac{\partial Q_x}{\partial t} + \frac{\partial (Q_x^2/A_x)}{\partial x} + gA_x \left(\frac{\partial H}{\partial x} + S_{fx} \right) = 0 \tag{2}$$

Where Q_x is the flowrate; x,t are spatial and temporal coordinates; A_x is the flow area; g is gravity; H is the water surface elevation; and S_{fx} is a friction slope. It is assumed that S_{fx} is approximated from Manning's equation for steady flow by (eg. [1])

$$Q_x = \frac{1.486}{n} A_x R^{2/3} S_{fx}^{1/2} \tag{3}$$

where R is the hydraulic radius; and n is a friction factor which may be increased to account for other energy losses such as expansions and bend losses. Letting m_x be a momentum quantity defined by

$$m_x = \left(\frac{\partial Q_x}{\partial t} + \frac{\partial (Q_x^2/A_x)}{\partial x} \right) \Big/ gA_x \tag{4}$$

then Eq. (2) can be rewritten as

$$S_{fx} = - \left(\frac{\partial H}{\partial x} + m_x \right) \tag{5}$$

In Eq. (4), the subscript x included in m_x indicates the directional term. The expansion of Eq. (2) to the two-dimensional case leads directly to the terms (m_x, m_y) except that now a cross-product of flow velocities is included, increasing the computational effort considerably.

Rewriting Eq. (3) and including Eqs. (4) and (5), the directional flow rate is computed by

$$Q_x = - K_x \left(\frac{\partial H}{\partial x} + m_x \right) \tag{6}$$

where Q_x indicates a directional term, and K_x is a type of conduction parameter defined by

$$K_x = \frac{1.486}{n} A_x R^{2/3} \Big/ \left| \frac{\partial H}{\partial x} + m_x \right|^{1/2} \tag{7}$$

In Eq. (7), K_x is limited in value by the denominator term being checked for a smallest allowable magnitude.

Substituting the flow rate formulation of Eq. (6) into Eq. (1) gives a diffusion type of relationship

$$\frac{\partial}{\partial x} K_x \left(\frac{\partial H}{\partial x} + m_x \right) = \frac{\partial A_x}{\partial t} \tag{8}$$

The one-dimensional diffusion model of Akan and Yen [1] assumes $m_x = 0$ in Eq. (7). Likewise, the two-dimensional diffusion model of Xanthopoulos and Koutitas [16] assumes $m_x = m_y = 0$. Thus, the one-dimensional diffusion model is given by

$$\frac{\partial}{\partial x} K_x \frac{\partial H}{\partial x} = \frac{\partial A_x}{\partial t} \tag{9}$$

where K_x is now simplified as

$$K_x = \frac{1.486}{n} A_x R^{2/3} \Big/ \left| \frac{\partial H}{\partial x} \right|^{1/2} \tag{10}$$

For a constant channel width, W, Eq. (9) reduces to

$$\frac{\partial}{\partial x} K_x \frac{\partial H}{\partial x} = W \frac{\partial H}{\partial t} \tag{11}$$

The remainder of this study considers the use of Eq. (8) as an alternative to the solution of the coupled system of Eqs. (1) and (2). However, it is noted that a family of models is given by Eq. (8) where m_x is defined by selecting from the possibilities

$$m_x = \begin{cases} \dfrac{\partial(Q_x^2/A_x)}{\partial x} \Big/ gA_x, & \text{(convective acceleration model)} \\ \dfrac{\partial Q_x}{\partial t} \Big/ gA_x, & \text{(local acceleration model)} \\ \left(\dfrac{\partial Q_x}{\partial t} + \dfrac{\partial(Q_x^2/A_x)}{\partial x}\right) \Big/ gA_x, & \text{(coupled model)} \\ 0\,, & \text{(diffusion model)} \end{cases} \tag{12}$$

NUMERICAL SOLUTION ALGORITHM

The following steps are taken in the computer program where the flow path is assumed initially discretized by equally spaced nodal points with a Mannings n, an elevation, and an initial flow depth (usually zero) defined:

(1) between nodal points, compute an average Mannings n, and average geometric factors
(2) assuming $m_x = 0$, estimate the nodal flow depths for the next timestep, $(t + \Delta t)$ by using Eqs. (9) and (10) explicitly
(3) using the flow depths at time t and $(t + \Delta t)$, estimate the mid-timestep value of m_x selected from Eq. (12)
(4) recalculate the conductivities K_x using the appropriate m_x values
(5) *determine the new nodal flow depths at time* $(t + \Delta t)$ using Eq. (8)
(6) return to Step (3) until K_x matches mid-timestep estimates.

The above algorithm steps can be used regardless of the choice of definition for m_x from Eq. (12). Additionally, the above program steps can be directly applied to a two-dimensional diffusion model with the selected (m_x, m_y) relations incorporated.

MODEL TIMESTEP SELECTION

The sensitivity of the model to timestep selection is dependent upon the slope of the hydrograph $(\delta Q/\delta t)$ and the grid spacing. Increasing the grid spacing size introduces additional storage to a corresponding increase in nodal point flood depth values. Similarly, a decrease in timestep size allows a refined calculation of inflow and outflow values and a smoother variation in nodal point depths with respect to time. The computer algorithm may self select a timestep by increments of halving (or doubling) the selected timestep size so that a proper balance of *inflow-outflow to control volume storage* variation is achieved. In order to avoid a matrix solution for flood depths, an explicit timestepping algorithm is used to solve for the time derivative term. For large timesteps or a rapid variation in the dam-break hydrograph (i.e., $(\delta Q/\delta t)$ large), a large accumulation of flow volume will occur at the most upstream nodal point. That is, at the dam-break reservoir nodal point, the lag in outflow from the control volume can cause unacceptable error in the computation of the flood depth. One method which offset this error is the program to self select the timestep until the difference in the rate of volume accumulation is within a specified tolerance. For the example problems considered, a timestep of about 5 to 10 *seconds was found adequate.*

Due to the form of the diffusion model in Eq. (11), the model can be extended into an implicit technique. However, this extension would require a matrix solution process which may become unmanageable for two-dimensional models which utilize hundreds of nodal points.

STUDY APPROACH

In order to evaluate the accuracy of the diffusion model of Eq. (11) in the prediction of flood depths, the fully dynamic flow model K-634 [7, 8] is used to determine channel *flood depths for comparison purposes. The K-634 model solves* the coupled flow equations of continuity and momentum by an implicit finite difference approach and is considered to be a highly accurate model for many unsteady flow problems. The study approach is to compare predicted flood depths predicted from both the K-634 and the diffusion (Eq. 11) model for various channel slopes and inflow hydrographs.

In this case study, two hydrographs are assumed, *namely, peak flows of 120,000 cfs and 600,000 cfs.* Both hydrographs are assumed to increase linearly from zero to the peak flow rate at time of 1-hour, and then decrease linearly to zero at time of 6-hours (See Fig. 1 insert). The study channel is assumed to be a uniform rectangular section of Manning's n equal to 0.040, and various slopes S_o in the range of 0.001 S_o 0.01. Figure 1 shows the comparison of modeling results. From the figure, various flood depths are plotted along the channel length of up to 10 miles. Two reaches of channel lengths of up to 30 miles are also plotted in Figure 1 which correspond to a slope $S_o = 0.0020$. In all tests, grid spacing was set at 1000-foot intervals.

From Figure 1, it is seen that the diffusion model provides estimates of flood depths that compare very well to the flood depths predicted from the K-634 model. Differences in predicted flood depths are less than 3-percent for the various channel slopes and peak flow rates considered.

GRID SPACING SELECTION

The choice of timestep and grid size for an explicit time advancement is a relative matter and is theoretically based on the well known Courant condition. The choice of grid size usually depends on available topographic data for nodal elevation determination and the size of the problem. The effect of the grid size (for constant timestep of 7.2 seconds) on the diffusion model accuracy can be shown by example where nodal spacings of 1000, 2000 and 5000 feet are considered. The predicted flood depths varied only slightly from choosing the grid size between 1000 feet and 2000 feet. However, an increased variation in results occurs when a grid size of 5000 feet is selected. Figure 2 shows the computed flood depths in comparison to the K-634 modeling results (Figure 1) for the considered grid sizes, and the peak flow rate *test hydrograph of 600,000 cfs.*

Because the algorithm presented is based upon an explicit timestepping technique, the modeling results may become inaccurate should the timestep size versus grid size ratio become large. A simple procedure to eliminate this instability is to halve the timestep size until convergence in computed results is achieved. Generally, such a timestep adjustment may be directly included in the dam-break model computer program. For the cases considered in this paper, timestep sizes of 7.2 seconds were found to be adequate when using the 1000-feet to 5000-feet grid sizes.

CONCLUSION

For the dam-break hydrographs considered and the range of channel slopes modeled, the simple diffusion dam-break model of Eq. (11) provides estimates of flood depths which compare favorably to the flood depths determined by the well known K-634 one-dimensional dam-break model. Generally speaking, the difference between the two modeling approaches is found to be less than a 3-percent variation in predicted flood depths.

The presented diffusion dam-break model is based upon a straightforward explicit timestepping method which allows the model to operate upon the nodal points without the need to use large matrix systems. Consequently, the model can be implemented on most currently available microcomputers.

The diffusion model of Eq. (11) can be directly extended to a two-dimensional model by adding the y-direction terms which are computed in a similar fashion as the x-direction terms. The resulting two-dimensional diffusion dam-break model is tested by modeling the considered test problems in the x-direction, the y-direction, and along a 45-degree trajectory across a two-dimensional grid aligned with the x-y coordinate axis. Using a similar two-dimensional model, Xanthopoulos and Koutitas [16] conceptually verify the diffusion modeling technique by considering the evolution of a two-dimensional flood plain which propagates radially from the dam-break site.

DISCUSSION

From the above conclusions, use of the diffusion approach of Eq. (11) in a two-dimensional dam-break model may be justifiable due to the low variation in predicted flooding depths (one-dimensional) with the exclusion of the inertia terms. Generally speaking, a two-dimensional model would be employed when the expansive nature of flood flows over a broad plain is anticipated. Otherwise, one of the available one-dimensional models would suffice for the analysis. Assuming a two-dimensional diffusion dam-break model is applied to such a broad plain, the deviation in predicted flood depths by not including the inertia terms may be considerably offset by the uncertainty associated with an assumed dam-break failure rate, and the assumed deterioration shape of the dam (e.g., [16]). An additional major uncertainty is the assumed friction factors assigned to the nodal points. Because such uncertainties remain with a predictive dam-break analysis, further refinement of the diffusion modeling approach by including the inertia terms may not be needed.

REFERENCES

[1] Akan, A.O., Yen, B.C., "Diffusion-Wave Flood Routing in Channel Networks," A.S.C.E. Hyd. Div., HY6, (1981).

[2] Chen, C., "Laboratory Verification of a Dam-Break Flood Model," A.S.C.E. Journ. of Hyd. Div., HY4, (1980).

[3] Chen, C., and Armbruster, J.T., "Dam-Break Wave Model: Formulation and Verification," A.S.C.E. Journ. of Hyd. Div., HY5, (1980).

[4] Henderson, F.M., "Flood Waves in Prismatic Channels," A.S.C.E., Journ. of Hyd. Div., No. HY4, (1953).

[5] Hunt, B., "Asymptotic Solution for Dam-Break Problem," A.S.C.E., Journ. Hyd. Div., HY1, (1982).

[6] Katopodes, N., and Strelkoff, T., "Computing Two-Dimensional Dam-Break Flood Waves," A.S.C.E., Journ. of Hyd. Div., HY9, (1978).

[7] Land, L.F., "Mathematical Simulations of the Toccoa Falls, Georgia, Dam-Break Flood," Wat. Res. Bul. (16), No. 6, (1980a).

[8] Land, L.F., "Evaluation of Selected Dam-Break Flood-Wave Models by Using Field Data," U.S.G.S. Wat. Res. Investigations, (1980b).

[9] Miller, W.A., and Cunge, J.A., "Simplified Equations of Unsteady Flow," Unsteady Flow in Open Channels, Water Res. Pub., Fort Collins, CO.

[10] Morris, E.M., and Woolhiser, D.A., "Unsteady One-Dimensional Flow Over a Plane: Partial Equilibrium and Recession Hydrographs, "Water Res. Res., AGU, Vol. 16, No. 2, (1980).

[11] Ponce, V.M., "Nature of Wave Attenuation in Open Channel Flow," A.S.C.E. Journ. of Hyd. Div., No. HY2, (1982).

[12] Ponce, V.M., Li, R.M., and Simons, D.B., "Applicability of Kinematic and Diffusion Models," Verification of Mathematical and Physical Models in Hydraulic Engineering, A.S.C.E. Hyd. Division, (1978).

[13] Ponce, V.M., and Tsivoglou, A.J., "Modeling Gradual Dam Breaches," A.S.C.E., Journ. Hyd. Div. HY7, (1981).

[14] Rajar, R., "Mathematical Simulation of Dam-Break Flow," A.S.C.E., Journ. Hyd. Div., HY7, (1978).

[15] Sakkas, J.G., and Strelkoff, T., "Dam-Break Flood in a Prismatic Dam Channel," A.S.C.E. Journ. of Hyd. Div., HY12, (1973).

[16] Xanthopoulos, T., and Koutitas, C., "Numerical Simulation of a Two-Dimensional Flood Wave Propagation Due to Dam Failure," Journ. of Hyd. Res. 14, No. 4, (1976).

FIG. 2
EFFECTS OF
DIFFERING
GRID SIZE





FIG. 1. DIFFUSION MODEL (⊙) AND K-634 MODEL RESULTS (SOLID LINE) FOR 1,000-FOOT WIDTH CHANNEL, MANNING'S n = 0.040, AND VARIOUS CHANNEL SLOPES, S_o.

APPENDIX

One-Dimensional Diffusion Dam-Break Model

DATA PREPARATION

Data is entered into file DAM1D.DAT and the program solutions are contained in file DAM1D.ANS. Data entry is as follows:

Line #	Variables
1	NNOD, DELT, TIME, TOUT, KOUT
2	B, XLEN, SO, CN
3	Z1
4	NUH
5	Q(I), HR(I)
.	.
.	.
5 + (NUH-1)	.
5 + NUH	JOUT.

DESCRIPTION OF VARIABLES

NNOD	=	Total number of nodes (evenly spaced)
DELT	=	Time interval (hours)
TIME	=	Total simulation time (hours)
TOUT	=	Output time interval (hours)
KOUT	=	1: Summary of maximum nodal depth values 0: Detailed results at each output interval
B	=	Rectangular channel width (feet)
XLEN	=	Constant distance between nodal points (feet)
SO	=	Slope of the channel bottom (ft/ft)
CN	=	Manning's friction coefficient
Z1	=	Channel bottom elevation at Node 1
NUH	=	Number of data pairs for inflow hydrograph
Q(I), HR(I)	=	Inflow rate in cfs and its corresponding time in hours
JOUT	=	1: Critical depth assumption for the downstream nodal point 0: Normal depth assumption for the downstream nodal point

PROGRAM LISTING

```
C
C    ONE DIMENSIONAL DIFFUSION DAM BREAK MODEL
C
       COMMON/BLK 1/ELE(100)
       COMMON/BLK 2/WAT(100)
       COMMON/BLK 3/Q(10),HR(10)
       COMMON/BLK 4/DMAX(100)
C
C    INPUT DATA
C
       NRD=1
       OPEN(UNIT=NRD,NAME='DAM1D.DAT',TYPE='OLD')
C...NODE NUMBER,TIME STEP,TOTAL SIMULATION TIME
C    AND OUTPUT INTERVAL
       READ(NRD,*)NNOD,DELT,TIME,TOUT,KOUT
C...CHANNEL INFORMATION: WIDTH,LENGTH OF ELEMENT,SLOPE,
C                        AND MANNING'S COEFFICIENT.
       READ(NRD,*)B,XLEN,SO,CN
C...CHANNEL ELEVATION FOR NODE #1
       READ(NRD,*)Z1
C...INFLOW HYDROGRAPH
       READ(NRD,*)NUH
       READ(NRD,*)((Q(I),HR(I)),I=1,NUH)
C...OUTFLOW CONDITION
C    JOUT=1 INDICATES CRITICAL DEPTH ASSUMPTION
C    JOUT=2 INDICATES NORMAL DEPTH ASSUMPTION
       READ(NRD,*)JOUT
C
       CLOSE(UNIT=NRD)
C
C    OUTPUT DATA
C
C...FORMATS
1      FORMAT(/,3X,'ONE DIMENSIONAL DIFFUSION DAM BREAK MODEL',
      1//,5X,'TIME STEP = ',F8.6,' HOUR',/,5X,'TOTAL SIMULATION',
      2' TIME = ',F8.2,' HOURS',/,5X,'TOTAL NODE NUMBER = ',I3,
      3/,5X,'ELEMENT LENGTH = ',F8.2,' FT')
2      FORMAT(5X,'CHANNEL WIDTH = ',F8.2,' FT',/,5X,'CHANNEL',
      1' SLOPE = ',F8.6,/,5X,'MANNING COEFFICIENT = ',F8.6)
3      FORMAT(5X,'INFLOW HYDROGRAPH :',/,6X,'INFLOW(CFS)   HOUR')
4      FORMAT(7X,F8.0,F8.2)
```

```
5      FORMAT(/,80('='),//)
6      FORMAT(/,5X,'TIME = ',F5.2,' HOUR',/,5X,
      1'NODE    DEPTH    ELEVATION')
7      FORMAT(5X,I3,2X,F8.4,3X,F8.2)
8      FORMAT(/,80('-'),/)
9      FORMAT(2X,'MAXIMUM DEPTH (FT)',/,2X,18('-'),/)
10     FORMAT(3X,'DEPTH',10F7.2,/)
11     FORMAT(4X,'NODE',3X,I3,9(4X,I3))
C
       NWT=2
       OPEN(UNIT=NWT,NAME='DAM1D.ANS',TYPE='NEW')
       WRITE(NWT,1)DELT,TIME,NNOD,XLEN
       WRITE(NWT,2)B,SO,CN
       WRITE(NWT,3)
       WRITE(NWT,4)((Q(I),HR(I)),I=1,NUH)
       WRITE(NWT,5)
C
C    COMPUTE THE CHANNEL ELEVATION, AND WATER SURFACE ELEVATION
C
       ELE(1)=71
       WAT(1)=ELE(1)
       DMAX(1)=0.
       DO 20 I=2,NNOD
       ELE(I)=ELE(I-1)-SO*XLEN
       WAT(I)=ELE(I)
       DMAX(I)=0.
20     CONTINUE
C
C    MAIN LOOP
C
       NIT=IFIX(TIME/DELT)
       IOUT=IFIX(TOUT/DELT)
       KK=0
       XTIME=0.
       DO 100 J=1,NIT
       KK=KK+1
       XTIME=XTIME+DELT
C...COMPUTE THE FLUX THROUGH THE FIRST CONTROL VOLUME
       GRADE=(WAT(2)-WAT(1))/XLEN
       DE=(WAT(2)-ELE(2)+WAT(1)-ELE(1))*.5
       CALL FLUX(GRADE,DE,B,QE,CN)
C...COMPUTE THE INFLOW RATE
       CALL INFLOW(XTIME,NUH,QIN)
C...COMPUTE THE NEW WATER SURFACE FOR NODE '1'
       WAT(1)=WAT(1)+(QIN-QE)*DELT*3600./(XLEN*B)
       IF(WAT(1).LT.ELE(1))WAT(1)=ELE(1)
       D1=WAT(1)-ELE(1)
       IF(D1.GT.DMAX(1))DMAX(1)=D1
       DO 30 I=2,NNOD-1
C...COMPUTE THE FLUX THROUGH THE WEST SIDE OF THE CONTROL VOLUME
       GRADW=(WAT(I)-WAT(I-1))/XLEN
       DW=(WAT(I)-ELE(I)+WAT(I-1)-ELE(I-1))*.5
       CALL FLUX(GRADW,DW,B,QW,CN)
C...COMPUTE THE FLUX THROUGH THE EAST SIDE OF THE CONTROL VOLUME
       GRADE=(WAT(I+1)-WAT(I))/XLEN
       DE=(WAT(I+1)-ELE(I+1)+WAT(I)-ELE(I))*.5
       CALL FLUX(GRADE,DE,B,QE,CN)
C...COMPUTE THE NEW WATER SURFACE FOR NODE 'I'
       WAT(I)=WAT(I)+(QW-QE)*DELT*3600./(XLEN*B)
       IF(WAT(I).LT.ELE(I))WAT(I)=ELE(I)
       DI=WAT(I)-ELE(I)
       IF(DI.GT.DMAX(I))DMAX(I)=DI
       IF(QW.EQ.0. .AND. QE.EQ.0.)GO TO 90
30     CONTINUE
C...COMPUTE THE FLUX THROUGH THE LAST CONTROL VOLUME
       GRADW=(WAT(NNOD)-WAT(NNOD-1))/XLEN
       DW=(WAT(NNOD-1)-ELE(NNOD-1)+WAT(NNOD)-ELE(NNOD))*.5
       CALL FLUX(GRADW,DW,B,QW,CN)
       DN=WAT(NNOD)-ELE(NNOD)
       IF(JOUT.EQ.1)GO TO 60
C...COMPUTE THE NEW WATER SURFACE BY NORMAL DEPTH ASSUMPTION
       IF(DN.GT.0.)CALL FLUX(SO,DN,B,QOUT,CN)
       IF(DN.LE.0.)QOUT=0.
       GO TO 70
```

```
C...COMPUTE THE NEW WATER SURFACE BY CRITICAL DEPTH ASSUMPTION
60    IF(DN.GT.0.)QOUT=5.6745*B*(DN)**1.5
      IF(DN.LE.0.)QOUT=0.
70    WAT(NNOD)=WAT(NNOD)+(QW-QOUT)*DELT*3600./(XLEN*B)
      IF(WAT(NNOD).LT.ELE(NNOD))WAT(NNOD)=ELE(NNOD)
      DL=WAT(NNOD)-ELE(NNOD)
      IF(DL.GT.DMAX(NNOD))DMAX(NNOD)=DL
C...OUTPUT THE RESULTS
90    IF(KK.NE.IOUT)GO TO 100
      KK=0
      IF(KOUT.EQ.1)GO TO 100
      WRITE(NWT,6)XTIME
      DO 40 I=1,NNOD
      DEPTH=WAT(I)-ELE(I)
      WRITE(NWT,7)I,DEPTH,ELE(I)
40    CONTINUE
      WRITE(NWT,8)
100   CONTINUE
      WRITE(NWT,9)
      K=1
      II=1
      JJ=10
200   WRITE(NWT,11)(J,J=II,JJ)
      WRITE(NWT,10)(DMAX(J),J=II,JJ)
      IF(JJ.EQ.NNOD)GO TO 300
      K=K+1
      II=II+10
      JJ=10*K
      IF(JJ.LT.NNOD)GO TO 200
      IF(JJ-NNOD.GT.10)GO TO 300
      JJ=NNOD
      GO TO 200
300   WRITE(NWT,5)
      CLOSE(UNIT=NWT)
      STOP
      END

C-----------------------------------------------------------------
C     SUBROUTINE FLUX
C-----------------------------------------------------------------
      SUBROUTINE FLUX(GRAD,D,B,Q,CN)
C
C     THIS SUBROUTINE COMPUTES THE MEAN FLUX THROUGH THE CONTROL VOLUME
C
      IF(D.LT.0.01)GO TO 10
      A=D*B
      RH=A/(2.*D+B)
      Q=1.486*A*(RH**0.6667)*SQRT(ABS(GRAD))/CN
      RETURN
10    Q=0.
      RETURN
      END
C-----------------------------------------------------------------
C     SUBROUTINE INFLOW
C-----------------------------------------------------------------
      SUBROUTINE INFLOW(TIME,NUH,QIN)
C
C     THIS SUBROUTINE USES A LINEAR INTERPOLATION FUNCTION
C     FOR INFLOW HYDROGARPH
C
      COMMON/BLK 3/Q(10),HR(10)
C
      IF(TIME.EQ.0.)GO TO 1000
      DO 100 I=2,NUH
      IF(TIME.GT.HR(I))GO TO 100
      QIN=Q(I-1)+(Q(I)-Q(I-1))*(TIME-HR(I-1))/(HR(I)-HR(I-1))
      GO TO 200
100   CONTINUE
1000  QIN=Q(1)
200   RETURN
      END
```

EXAMPLE PROBLEM INPUT

```
80 .002 6 1 1
1000 1000 .010 .04
1000
3
0 0
120000 1
0 6
0
```

EXAMPLE PROBLEM RESULTS

```
ONE DIMENSIONAL DIFFUSION DAM BREAK MODEL

  TIME STEP = 0.002000 HOUR
  TOTAL SIMULATION TIME =       6.00 HOURS
  TOTAL NODE NUMBER =  80
  ELEMENT LENGTH =  1000.00 FT
  CHANNEL WIDTH =  1000.00 FT
  CHANNEL SLOPE = 0.004000
  MANNING COEFFICIENT = 0.040000
  INFLOW HYDROGRAPH :
   INFLOW(CFS)    HOUR
          0.      0.00
     120000.      1.00
          0.      6.00

==============================================

  TIME =   1.00 HOUR
  NODE    DEPTH    ELEVATION
    1    10.5581    1000.00
    2    10.4755     996.00
    3    10.3911     992.00
    4    10.3050     988.00
    5    10.2170     984.00
    6    10.1271     980.00
    7    10.0351     976.00
    8     9.9409     972.00
    9     9.8444     968.00
   10     9.7454     964.00
   11     9.6436     960.00
   12     9.5389     956.00
   13     9.4311     952.00
   14     9.3198     948.00
   15     9.2048     944.00
   16     9.0857     940.00
   17     8.9623     936.00
   18     8.8345     932.00
   19     8.7031     928.00
   20     8.5704     924.00
   21     8.4425     920.00
   22     8.3347     916.00
   23     8.2785     912.00
   24     7.8172     908.00
   25     3.1645     904.00
   26     0.2136     900.00
   27     0.0011     896.00
   28     0.0000     892.00
   29     0.0000     888.00
   30     0.0000     884.00
```

31	0.0000	880.00
32	0.0000	876.00
33	0.0000	872.00
34	0.0000	868.00
35	0.0000	864.00
36	0.0000	860.00
37	0.0000	856.00
38	0.0000	852.00
39	0.0000	848.00
40	0.0000	844.00
41	0.0000	840.00
42	0.0000	836.00
43	0.0000	832.00
44	0.0000	828.00
45	0.0000	824.00
46	0.0000	820.00
47	0.0000	816.00
48	0.0000	812.00
49	0.0000	808.00
50	0.0000	804.00
51	0.0000	800.00
52	0.0000	796.00
53	0.0000	792.00
54	0.0000	788.00
55	0.0000	784.00
56	0.0000	780.00
57	0.0000	776.00
58	0.0000	772.00
59	0.0000	768.00
60	0.0000	764.00
61	0.0000	760.00
62	0.0000	756.00
63	0.0000	752.00
64	0.0000	748.00
65	0.0000	744.00
66	0.0000	740.00
67	0.0000	736.00
68	0.0000	732.00
69	0.0000	728.00
70	0.0000	724.00
71	0.0000	720.00
72	0.0000	716.00
73	0.0000	712.00
74	0.0000	708.00
75	0.0000	704.00
76	0.0000	700.00
77	0.0000	696.00
78	0.0000	692.00
79	0.0000	688.00
80	0.0000	684.00

MAXIMUM DEPTH (FT)

NODE	1	2	3	4	5	6	7	8	9
DEPTH	10.63	10.62	10.61	10.60	10.59	10.58	10.58	10.57	10.56
NODE	11	12	13	14	15	16	17	18	19
DEPTH	10.55	10.55	10.54	10.54	10.53	10.53	10.52	10.52	10.52
NODE	21	22	23	24	25	26	27	28	29
DEPTH	10.51	10.50	10.50	10.50	10.49	10.49	10.49	10.48	10.48
NODE	31	32	33	34	35	36	37	38	39
DEPTH	10.47	10.47	10.47	10.46	10.46	10.46	10.45	10.45	10.45
NODE	41	42	43	44	45	46	47	48	49
DEPTH	10.44	10.44	10.43	10.43	10.42	10.42	10.41	10.41	10.40
NODE	51	52	53	54	55	56	57	58	59
DEPTH	10.39	10.38	10.37	10.36	10.36	10.35	10.34	10.33	10.32
NODE	61	62	63	64	65	66	67	68	69
DEPTH	10.29	10.28	10.27	10.26	10.25	10.24	10.22	10.21	10.20
NODE	71	72	73	74	75	76	77	78	79
DEPTH	10.17	10.16	10.15	10.14	10.12	10.11	10.10	10.08	10.07